
Wren est un langage de script assez simple, basé sur des classes et destiné à être embarqué dans des applications (notamment des jeux vidéos). J'avais envie d'essayer ce langage depuis un moment, je me suis dit qu'il fallait participer à la hype du moment.


Donc, suite au portage de TapTempo en divers langage: Rust, Ada, JavaScript, Perl, Python 2.7, Bash, Perl6 Haskell et PWA, voici la version en Wren.


import "io" for Stdin, Stdout

class TapTempo {
  construct new() {
    _sampleSize = 5
    _resetTime = 5.0
    _precision = 0
    _hitTimePoints = []
  }

  computeBpm(currentTime, lastTime, occurenceCount) {
    if (occurenceCount == 0) {
      occurenceCount = 1
    }

    var elapsedTime = currentTime - lastTime

    System.print("elapsedTime: %(elapsedTime)")

    var meanTime = elapsedTime / occurenceCount
    var bpm = 60.0 / meanTime

    return bpm
  }

  isResetTimeElapsed(currentTime, lastTime) {
    var elapsedTime = currentTime - lastTime
    return elapsedTime > _resetTime
  }

  run() {
    System.print("Hit enter key for each beat (q to quit).")

    var shouldContinue = true

    while (shouldContinue) {
      var i = "q"

      while (i.count != 0) {
        i = Stdin.readLine()

        if (i == "q") {
          shouldContinue = false
          System.print("Bye Bye!")
          break
        }
      }

      if (shouldContinue) {
        var currentTime = System.clock

        System.print("currentTime %(currentTime) %(System.clock)")

        if (_hitTimePoints.count > 0 && isResetTimeElapsed(currentTime, _hitTimePoints[_hitTimePoints.count - 1])) {
          _hitTimePoints.clear()
        }

        _hitTimePoints.add(currentTime)

        if (_hitTimePoints.count > 1) {
          System.write("\r")

          var bpm = computeBpm(_hitTimePoints[_hitTimePoints.count - 1], _hitTimePoints[0], _hitTimePoints.count - 1)
          System.write("Tempo: %(bpm.round) bpm\t")
        } else {
          System.write("\r")
          System.write("[Hit enter key one more time to start bpm computation...]")
        }

        Stdout.flush()

        while (_hitTimePoints.count > _sampleSize) {
          _hitTimePoints.removeAt(0)
        }

      }
    }

  }
}

var tapTempo = TapTempo.new()
tapTempo.run()



À noter que cette version peut donner des résultats bizarres. En effet, System.clock est basé sur clock(3) qui fournit le temps CPU, et pas le temps global. Donc quand on a un gros processus qui tourne et qui bouffe le CPU, on peut avoir des temps fourni par cet appel qui ne correspondent pas vraiment à ce qu'on recherche. Mais dans l'idée, le code est juste et essaie de se rapprocher le plus possible de l'original, la gestion des options en moins.

